

IF AND WHERE PHRASES

Basic IF and WHERE commands perform the same functions, though syntax requirements like the use of single quotes and parentheses differ, the results are the same.

Why is the WHERE phrase important? It eliminates the need for logical defines, better known as IF...THEN...ELSE. As a result, your reports will be easier to write, will process more efficiently, and will cost less money.

Following are examples of basic IF and WHERE phrases:

IF FIELDNAME OPERATOR VALUE	WHERE FIELDNAME OPERATOR VALUE
IF AGY EQ 100	WHERE AGY IS '100'
IF UNIT EQ 210 OR 320	WHERE UNIT IS '210' OR '320'
IF SALTOT GT 4000	WHERE SALTOT EXCEEDS 4000
IF STATCD NE 'PERM SEP'	WHERE STATCD IS-NOT 'PERM SEP'

Tips to remember:

- In the examples above, we've used the operators EQ and IS; GT and EXCEEDS, NE and IS-NOT. These commands can be used interchangeably. See your FOCUS Manual for more information on operators.
- In WHERE phrases, alphanumeric and date values must be enclosed in single quotes.
- Parentheses are needed when using both AND and OR together.
- If you have reports that use IF statements you're getting the results that you need, you do not need to convert commands to WHERE phrases.

Next, we'll compare using define fields versus WHERE phrases in the sample reports listed below.

SAMPLE REPORT 1

We need a report that lists all employees who receive Bilingual Pay. Using a define, we have created a new field, called BILINGUAL, to determine whether an employee receives Bilingual Pay. Based on this define, each record will be assigned a value of YES or NO.

```
DEFINE FILE CSEMP1 ADD
BILINGUAL/A3=IF (EARNID1 EQ '8B' OR '8B1' OR '8B2' OR '8B3' ) OR
  (EARNID2 EQ '8B' OR '8B1' OR '8B2' OR '8B3') OR
  (EARNID3 EQ '8B' OR '8B1' OR '8B2' OR '8B3') THEN 'YES' ELSE 'NO';
END
```

Next, we will use the BILINGUAL field in a report to select only those records with a value of YES.

```
TABLE FILE CSEMP1
PRINT FULL_NAME AS 'NAME'
BY AGY AS 'AGENCY'
BY UNIT
IF BILINGUAL EQ YES (can also say WHERE BILINGUAL IS 'YES')
IF STATCD EQ ACTIVE
END
```

An easier way to accomplish the same result is to simply include a WHERE phrase in the report, to select only those employees receiving Bilingual Pay.

```
TABLE FILE CSEMP1
PRINT FULL_NAME AS 'NAME'
BY AGY AS 'AGENCY'
BY UNIT
WHERE ((EARNID1 IS '8B' OR '8B1' OR '8B2' OR '8B3') OR
       (EARNID2 IS '8B' OR '8B1' OR '8B2' OR '8B3') OR
       (EARNID3 IS '8B' OR '8B1' OR '8B2' OR '8B3'))
WHERE STATCD IS 'ACTIVE'
END
```

SAMPLE REPORT 2

We need a report that counts active, full time, managerial and supervisory employees and those designated E48, E59, E98 or E99. Using a define, we have created a new field, called **TEST** to identify these employees.

```
DEFINE FILE CSEMP1 ADD
TEST/A3=IF (EMPDSGN EQ 'M' OR 'S') OR (EMPCBID EQ 'E48' OR 'E59' OR
'E98' OR 'E99') THEN 'YES' ELSE 'NO';
END
```

Next, we will use the TEST field in a report to select only those records with a YES value.

```
TABLE FILE CSEMP1
COUNT SSN
BY DSGN_NAME AS 'BARGAINING, DESIGNATION'
IF TEST EQ YES (Can also say WHERE TEST IS 'YES')
IF STATCD IS ACTIVE
IF TBASE IS FT
END
```

An easier way to accomplish the same results, is to simply include a WHERE phrase in the report.

```
TABLE FILE CSEMP1
COUNT SSN
BY DSGN_NAME AS 'BARGAINING,DESIGNATION'
WHERE ((EMPDSGN IS 'M' OR 'S' ) OR
       (EMPCBID IS 'E48' OR 'E59' OR 'E98' OR 'E99'))
WHERE STATCD IS 'ACTIVE'
WHERE TBASE IS 'FT'
END
```